

D03 - 2008

Creating Special Symbols in SAS® Graph

Shiqun (Stan) Li, Minimax Information Services, NJ
Wei Zhou, Eli Lilly and Company, IN

This paper will present several techniques to embed special characters and special symbols into SAS® graphs. The special symbols can be Greek letters, mathematical symbols, subscription, superscription, underline, and user designed symbols. The symbols can be created in the titles, footnotes, axis labels, or the graph area of a SAS graph. This presentation is prepared for an intermediate and advanced audience.

Key Words: Special Symbol, Special Character, SAS/Graph, NOTE-MOVE, Subscription, Superscription, Underline, Customized Font.

Introduction:

Very often, we need to create special characters or special symbols in a SAS® output. If the SAS output is a listing or a table, it is not difficult to accomplish this. One simple practice is to map a special character into its Unicode and utilize SAS ODS to produce a RTF document. After that, the RTF file will automatically convert the Unicode into the corresponding special character or special symbol. There are several interesting papers discussing this SAS ODS/RTF technique^[1, 2]. However, when the SAS output is a graph, for example a plot, bar, or pie chart, etc., the SAS ODS/RTF method may not work. Alternative approaches are usually required.

In the following sections, we will demonstrate several tips to generate special symbols in a graph.

Special Symbols:

It is quite often that we need to use some special symbols, such as P-hat or X-bar in axis labels or in titles. To achieve this, we can use the SAS NOTE/MOVE function. An example to generate a P-hat in y-axis can be as following:

```
NOTE font=swiss move=(5,40)pct h=1.5 a=90 'P' move=(-1, -1.4) "^" ;
```

This NOTE statement tells SAS to generate a 'P' at the (x, y) coordinate position (5, 40), then move back (-1, -1.4) unit and add a "^". This will complete a P-hat in y-axis. A NOTE statement is usually used together with another axis or title statement. At the end, we will present an integrated example of using NOTE statement and the techniques followed.

NOTE-MOVE statement is a very unique and useful statement. It can be used to perform some functions done by SAS/Annotate, such as adding symbols or labels to a graph^[3]. It can also be

employed to draw underline or strikethrough as well as subscription and/or superscription^[4]. The NOTE statement below will produce a P₁-hat symbol:

```
NOTE f=swiss m=(5,40)pct h=1.5 a=90 'P' m=(-1,-1.4) "^" m=(+1.5,-.5) h=.75 '1';
```

A statement like this:

```
footnote1 h=1 j=L " This is only an example." move=(-16,+0) " ____ " ;
```

will produce a footnote in a graph like this: **This is only an example.**

Special Characters:

There are different methods to produce special characters, such as Greek letters or math symbols in a SAS graph. The usage of special graph fonts can be found from a SAS/Graph manual^[5]. In the TITLE statement below, we use three different graph fonts to generate a graph title as **SAS[®] Graph Example¹**

```
Title h=2 f=swissb "SAS" m=(+.5, +1) h=1.2 f=special "R"  
m=(+0, -1) h=2 f=swissb " Graph Example" m=(+.5,+0) f=century "%sysfunc(byte(185))";
```

There are various fonts available in SAS system. SAS/Graph document^[5] has the list of all SAS fonts. To preview the fonts/symbols and its corresponding character codes, we can use the following code:

```
proc gfont name=greek nobuild height=4  
romcol=Blue romfont=swissb romht=3 showroman;  
run; quit;
```

Fonts Designed by User:

Occasionally, users may need some very unique symbols. Those symbols may not be provided by SAS Institution. In such situations, SAS procedure PROC GFONT can be used to build customized 'fonts' or symbols^[5, 6]. A simple example of making a font is shown below. This Proc GFONT will build a symbol of a circle with lower one-third filled with a specific color.

```
data myFont;  
input char $ seg x y lp $ ptype $ ;  
datalines;
```

A	1	1	0	L	V
A	1	0	0	L	C
A	1	0	1	L	V
A	1	.	.	L	V
A	2	0	1	L	V

```

A 2 0 0 L C
A 2 -1 0 L V
A 2 . . L V
A 3 -1 0 L V
A 3 0 0 L C
A 3 -.87 -.5 L V
A 3 . . L V
A 4 0 -.5 P V
A 4 -.87 -.5 P V
A 4 0 0 P C
A 4 0 -1 P V
A 4 . . P V
A 5 0 -.5 P V
A 5 0 -1 P V
A 5 0 0 P C
A 5 .87 -.5 P V
A 5 . . P V
A 6 .87 -.5 L V
A 6 0 0 L C
A 6 1 0 L V
A 6 . . L V

```

<< part of data omitted >>

```

;
%let WorkDir = %sysfunc(pathname(WORK));
libname GFont0 "&WorkDir.";
Proc gFont data=myFont name=myFonts
    filled resolution=3 showroman;
run; quit;

```

Integrated Example:

In this example, we will integrate the techniques introduced above to create a demonstrating SAS graph. This sample code utilizes the four user-defined fonts ('A', 'B', 'C', 'D') built in the PROC GFONT illustration above. The output graph will be showed after the codes.

```

* 1) make-up a test dataset;
*** 3 time points, w/ 10 measures for each time point;
data test;
do t=1 to 3;
    do i=1 to 10; y=ranuni(0); age=ranuni(0)*100; output; end;
end;
run;
proc format;
    value tfmt 1="Q" 2="R" 3="QSR";
    value AgeGrp 0-<25=1 25-<50=2 50-<75=3 75-100=4;

```

```

        value $AgeLabl '1'='0-<25' '2'='25-<50' '3'='50-<75' '4'='75-100';
run;
data test;
set test;
  AgeGrp=put(age,AgeGrp.);
run;

* 2) specify the axes and graph symbols;
options reset=all ;
axis1 label=(h=1 pct ' ') value=(h=2 f=marker) offset=(5, 5) minor=none;
axis2 label=(a=90 h=7 pct ' ') minor=none;
axis3 label=none value=none major=none minor=none;

symbol1 v=none i=std2mj c=blue l=1; *** 2 std errors;
symbol2 v='A' f=myFonts h=.8 c=green;
symbol3 v='B' f=myFonts h=.8 c=blue;
symbol4 v='C' f=myFonts h=.8 c=purple;
symbol5 v='D' f=myFonts h=.8 c=black;

legend1 position=(top right inside) mode=protect label=("Age Group:" f=swissb)
        value=(f=swissb);

* 3) plot the graph;
proc gplot data=test;
plot y*t / haxis=axis1 vaxis=axis2 ;
plot2 y*t=AgeGrp / vaxis=axis3 legend=legend1;
format t tfmt3. AgeGrp $AgeLabl10.;

*** Y-axis:  $\hat{p}_1(\mu \pm 2\sigma)$  ;
*** Be sure to have the +/- in the m-vectors;
note f=swiss m=(5,40)pct h=1.5 a=90 'P' m=(-1,-1.4) "^" m=(+1.5,-.5) h=.75 '1'
        m=(-1.5, +.5) h=1.5 f=greek ' (m' f=math 'G' f=greek '2s' f=swiss ' )';


*** Special symbol in graph area: *
note f=swiss m=(13, 79)pct h=1.3 f=marker 'T' m=(+34, 77)pct h=1.3 f=marker 'TT'
        m=(+34, +2)pct h=1.3 f=marker 'T' ;

*** a title: SAS® Graph Example1 ;
title1 h=2 f=swissb "SAS" m=(+.5, +1) h=1.2 f=special "R"
        m=(+0, -1) h=2 f=swissb " Graph Example" m=(+.5,+0) f=century "%sysfunc(byte(185))";

*** a footnote: ** P value <= 0.01 ;
footnote1 h=1 j=L f=marker " TT" f=swissb " P value <= " underlin=1 '0.01';
footnote3 h=1 j=L f=marker " T" f=swiss " P value <= 0.05" move=(-3, +0) " ____" ;
run; quit;

```

SAS[®] Graph Example ¹



This graph was generated in SAS System for Windows, version 9.1.3.

Conclusion:

Special characters and special symbols can be created on SAS graphs via various techniques. In addition to the fonts and symbols provided by SAS Institute, users can create self-designed symbols with Proc GFont. With some enhanced fonts and selective symbols on a graph, the reports can be more customized and more self-explained.

References:

1. David Mink and David J. Pasta; Improving Your Graphics Using SAS/GRAPH[®] Annotate Facility, SUGI 31, <http://www2.sas.com/proceedings/sugi31/085-31.pdf>
2. Richard Rowell and Jim Lenihan; MORE CUSTOMIZATION?: CREATING SYMBOLS IN RTF FILES USING ODS; PharmaSug 2004, <http://www.lexjansen.com/pharmasug/2004/technicaltechniques/tt12.pdf>
3. Archer R. Gravely; Achieving Graphical Excellence With SAS/GRAPH; SUGI 25, <http://www2.sas.com/proceedings/sugi25/25/iv/25p164.pdf>

4. Lori D. Griffin; Graphing: Taking the Mystery out of Subscripts and Superscripts, SUGI 22, <http://www2.sas.com/proceedings/sugi22/INFOVIS/PAPER172.PDF>
5. SAS Manuals on SAS Graph Fonts; <http://www2.stat.unibo.it/ManualiSas/gref/c06.pdf>
6. SAS Manuals on Proc GFONT; <https://techsupport.cob.rit.edu/manuals/sas8/gref/zatement.htm>

TRADEMARKS:

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Contact the authors at:

Shiqun (Stan) Li

Minimax Information Services
49 Hendrickson Dr.
Belle Mead, NJ 08502
(908) 240-8229
shiqun@gmail.com

Wei Zhou

Eli Lilly and Company
Lilly Corporate Center
Indianapolis, IN 46285
Drop Code: 4133
(317) 651-0044
zhouwe@lilly.com